Assignment 5 Question 3

Part (a)

```
> f1,f2 := x*y-z,x*z-y;
```

$$f1, f2 := x\,y - z, x\,z - y \quad (1)$$

```
> with(Groebner):
> S12 := expand( z*f1-y*f2 );
```

$$S12 := y^2 - z^2 \quad (2)$$

Since LT(f1)=xy does not divide y^2 and LT(f2)=xz does not divide y^2 {f1,f2} is not a GB. Check

```
> R := NormalForm(S12,[f1,f2],plex(x,y,z));
```

$$R := y^2 - z^2 \quad (3)$$

```
> f[1],f[2],f[3] := x+1,x*y+1,y-1;
```

$$f_1, f_2, f_3 := x + 1, x\,y + 1, y - 1 \quad (4)$$

```
> for i to 3 do for j from i+1 to 3 do
      S[i,j] := SPolynomial(f[i],f[j],plex(x,y));
      R[i,j] := NormalForm(S[i,j],[f[1],f[2],f[3]],plex(x,y));
      print(i,j,S[i,j],R[i,j]);
  od od:
```

$$1, 2, y - 1, 0$$
$$1, 3, x + y, 0$$
$$2, 3, x + 1, 0 \quad (5)$$

So G = {f1,f2,f3} is a GB by Buchberger's S-polynomial criterion.

Part (b)

```
> f1,f2 := x-z^2,y-z^3;
```

$$f1, f2 := -z^2 + x, -z^3 + y \quad (6)$$

```
> S12 := SPolynomial(f1,f2,plex(x,y,z));
```

$$S12 := x\,z^3 - y\,z^2 \quad (7)$$

```
> NormalForm(S12,[f1,f2],plex(x,y,z));
```

$$0 \quad (8)$$

```
> G := [f1,f2];
```

$$G := [-z^2 + x, -z^3 + y] \quad (9)$$

So G = {f1,f2} is already a GB in >lex with x>y. Notice LT(f1)=x and LT(f2)=y are relatively prime.

```
> S12 := SPolynomial(f1,f2,grlex(x,y,z));
```

$$S12 := -x\,z + y \quad (10)$$

```
> f3 := NormalForm(S12,G,grlex(x,y,z));
```

$$f3 := -x\,z + y \quad (11)$$

```
> G := [f1,f2,f3];
```

$$G := [-z^2 + x, -z^3 + y, -x\,z + y] \quad (12)$$

```
> S13 := SPolynomial(f1,f3,grlex(x,y,z)):
  f4 := NormalForm(S13,G,grlex(x,y,z));
```

$$f4 := -x^2 + y\,z \tag{13}$$

```
> S23 := SPolynomial(f2,f3,grlex(x,y,z));
  NormalForm(S23,G,grlex(x,y,z));
```

$$S23 := y\,z^2 - x\,y$$

$$0 \tag{14}$$

```
> G := [f1,f2,f3,f4];
```

$$G := [-z^2 + x, -z^3 + y, -x\,z + y, -x^2 + y\,z] \tag{15}$$

```
> for i to 3 do S := SPolynomial(G[i],f4,grlex(x,y,z)); R :=
  NormalForm(S,G,grlex(x,y,z)); od;
```

$$S := y\,z^3 - x^3$$

$$R := 0$$

$$S := y\,z^4 - x^2\,y$$

$$R := 0$$

$$S := y\,z^2 - x\,y$$

$$R := 0 \tag{16}$$

So G is a GB and since

```
> map(LeadingMonomial,G,grlex(x,y,z));
```

$$[z^2, z^3, x\,z, x^2] \tag{17}$$

we can discart G[2] so

```
> G := [f1,f3,f4];
```

$$G := [-z^2 + x, -x\,z + y, -x^2 + y\,z] \tag{18}$$

```
> G := -G;
```

$$G := [z^2 - x, x\,z - y, x^2 - y\,z] \tag{19}$$

is the reduced GB

```
> Groebner[Basis]([f1,f2],grlex(x,y,z));
```

$$[z^2 - x, x\,z - y, x^2 - y\,z] \tag{20}$$

Part (c)

```
> f1,f2,f3 := x*y-1,x*z-1,y*z-1;
```

$$f1, f2, f3 := x\,y - 1, x\,z - 1, y\,z - 1 \tag{21}$$

```
> G := [f1,f2,f3];
```

$$G := [x\,y - 1, x\,z - 1, y\,z - 1] \tag{22}$$

```
> n := nops(G);
  R := {seq(seq(NormalForm(SPolynomial(G[i],G[j],grlex(x,y,z)),G,
  grlex(x,y,z)),j=i+1..n),i=1..n)};
```

$$n := 3$$

$$R := \{x - y, x - z, y - z\} \tag{23}$$

```
> G := [op(G),op(R)];
```

$$G := [x\,y - 1, x\,z - 1, y\,z - 1, x - y, x - z, y - z] \tag{24}$$

```
> n := nops(G);
```

```
R := {seq(seq(NormalForm(SPolynomial(G[i],G[j],grlex(x,y,z)),G,
grlex(x,y,z)),j=i+1..n),i=1..n)};
```

$$n := 6$$

$$R := \{0, z^2 - 1\} \quad (25)$$

```
> G := [op(G),R[2]];
```

$$G := [x y - 1, x z - 1, y z - 1, x - y, x - z, y - z, z^2 - 1] \quad (26)$$

```
> n := nops(G);
R := {seq(seq(NormalForm(SPolynomial(G[i],G[j],grlex(x,y,z)),G,
grlex(x,y,z)),j=i+1..n),i=1..n)};
```

$$n := 7$$

$$R := \{0\} \quad (27)$$

We have a GB. Let's make it minimal

```
> G := [G[4],G[6],G[7]];
```

$$G := [x - y, y - z, z^2 - 1] \quad (28)$$

is a minimal GB.

```
> map(LeadingMonomial,G,grlex(x,y,z));
```

$$[x, y, z^2] \quad (29)$$

So the terms in the remainder are linear combinations of 1,z.